

# ЛЕКЦИЯ 8

доц. д-р Стела Стефанова

## Входен език на Pspice A/D. Описание на цифрови стимул сигнали




### 1. Понятие за цифрови стимул сигнали

- Всички входни, захранващи, управляващи и установяващи сигнали, необходими за правилната работа на цифровата схема.

### 2. Видове цифрови стимул сигнали


Генераторите на цифрови стимул сигнали се делят на три групи в зависимост от начина на дефинирането им и са показани в Таблица 1.

Таблица 1. Видове цифрови стимул сигнали

ГРУПА	Име на стимул сигналите	Тип на стимул сигналите	Дефинират се в:	Библиотеки и графични изображения
<b>Група 1</b> (на базата на Stimulus Editor)	DIGSTM 1 DIGSTM 2 DIGSTM 4 DIGSTM 8 DIGSTM 16 DIGSTM 32	Единичен сигнал или магистрала	Редактор на стимул сигнали (Stimulus Editor)	<b>SOURCSTM.OLB</b>  Implementation =
<b>Група 2</b> (на базата на Property Editor)	DIGCLOCK STIM 1 STIM 4 STIM 8 STIM 16	Тактов сигнал Единичен сигнал 4-битов сигнал 8- битов сигнал 16- битов сигнал	Редактор на свойствата (Property Editor)	<b>SOURCE.OLB</b> 
<b>Група 3</b> (на базата на файл)	FILESTIM 1 FILESTIM 2 FILESTIM 4 FILESTIM 8 FILESTIM 16 FILESTIM 32	1- битов сигнал 2- битов сигнал 4- битов сигнал 8- битов сигнал 16- битов сигнал 32- битов сигнал	Редактор на свойствата на базата на предварително дефиниран файл	<b>SOURCE.OLB</b>  FILENAME = SIGNAME =


На Фиг. 1 са показани графичните изображения и библиотеките в OrCAD Capture CIS за различните групи цифрови стимул сигнали.

## DIGITAL STIMULI IN PSPICE A/D



Фиг. 1. Графичните изображения и библиотеки в OrCAD Capture CIS за групите цифрови стимул сигнали

### 3. Общ формат на описание на цифровите стимул сигнали



### 3.1. 1 Тип на цифровото устройство

- **STIM** – служебна дума, указваща типа на цифровото устройство т.е. стимул сигнал.

### 3.2. 2 Брой сигнали и формат

#### <Брой сигнали>

- Определя броя на изходите на цифровия генератор на сигнали;
- Съответства на свойство **WIDTH** в Property Editor;

TIMESTEP	Value	WIDTH
	STIM1	1

- Стойност по подразбиране = 1.

#### <Формат>

- Спецификация на начина, по който се представят логическите нива на сигнала от цифровия генератор;
- Тази променлива представлява последователност от цифри, сумата на които е равна на броя на сигналите;
- Всяка една от тези цифри може да бъде със стойност 1, 3 и 4, което отговаря на представянето на логическите нива в съответна бройна система:
  - 1 – двоично представяне (binary) =  $2^1$ .
  - 3 – осмично представяне (octal) =  $2^3$ .
  - 4 – шестнадесетично представяне (hex) =  $2^4$ .
- Съответства на свойство **FORMAT** в Property Editor;

DIG_GND	DIG_PWR	FORMAT
\$G_DGND	\$G_DPWR	1

- Стойност по подразбиране = 1;
- **Примери за дефиниране на параметрите брой сигнали и формат:**
  - U1 STIM (1, 1)** – цифров генератор на сигнали с един изход в двоичен формат;
  - U2 STIM (3, 111)** – три изхода в двоичен формат;
  - U3 STIM (16, 4444)** – 16 изхода, представени в шестнадесетичен формат. Сумата на цифрите 4+4+4+4 трябва да бъде равна на броя на изходите на цифровия генератор (16 сигнала).

### 3.3. 3 Плюс възел на захранващия източник и минус възел на захранващия източник

- Глобални възли за цифрово захранване и цифрова земя **\$G\_PWR** и **\$G\_DGND**.

### 3.4. 4 Списък от възли

- Указва се номера на възлите, в които са включени изходите на цифровия генератор;
- Броят на възлите трябва да бъде равен на стойността на променливата **<брой сигнали>**;
- **Пример: U3 STIM (3,111) \$g\_dpwr \$g\_dgnd**  
+ **INT RUN MODE** (имена на възли, в които са включени трите изхода).

### 3.5. 5 Име на модела на входа /изхода

- **IO\_STM** - служебна дума, указваща името на модела на входа /изхода на цифровия генератор.

### 3.6. 6 Степен на сложност на макромодела на А/Ц и Ц/А интерфейс - (IO\_LEVEL)

- Незадължителен параметър;
- Дефинира нивото на сложност на макромодела на А/Ц и Ц/А интерфейс;
- 4 нива на сложност на макромодела на А/Ц и Ц/А интерфейс;
- Стойност по подразбиране = 0 (**IO\_LEVEL = 0**).

### 3.7. 7 Име на стимул сигнала (STIMULUS)

- Незадължителен параметър;
- Служебна дума, след която се задава името на стимул сигнала;
- **Пример: STIMULUS = <име на сигнала>**

### 3.8. 8 Стъпка по времето (TIMESTEP)

- незадължителен параметър;
- дефинира равномерна стъпка във времето -  $\Delta t$ ;
- стойност по подразбиране = 0 (TIMESTEP = 0).

При описание на цифровите стимул сигнали се използват три начина за представяне на сигналите във времето – относителна форма, абсолютна форма и с цикъл.

#### 3.8.1. Относителна форма за дефиниране на времето


- текущият момент от време се дефинира по отношение на предишния момент, както е показано на Фиг. 2;
- за означението на този режим се използва символът "+" в позицията преди числената стойност;

##### Пример:

Описание на стимул сигнала от Фиг. 2

Time	Logical Level
0	0
+100n	1
+100n	0
+100n	1
+100n	1
+100n	0

- **Описание на стимул сигнала:** В нулевия момент сигналът е с логическо ниво 0, след 100 ns логическото ниво става 1, след още 100 ns логическото ниво е 0 и т.н.



Фиг. 2. Относителна форма на дефиниране на времето

#### 3.8.2. Дефиниране на времето с брой цикли

- текущият момент от време се дефинира с номера на цикъла, умножен по равномерната стъпка по времето, както е показано на Фиг. 3:

$$t_i = i \cdot \text{timestep}$$


- използва се параметърът **TIMESTEP** за дефиниране на големината на стъпката във времето;
- работи се с равномерна стъпка във времето т.е. стъпка с една и съща стойност;
- използва се суфикс "c" за означение на цикъл във времето (cycle);

##### Пример:

Описание на стимул сигнала от Фиг. 3:

Time	Logical Level
0c	0
1c	1
2c	0
3c	1

- **Описание на стимул сигнала:** В нулевия цикъл сигналът е с логическо ниво 0, в първия цикъл е с логическото ниво 1, във втория цикъл е с логическото ниво 0 и т.н.



Фиг 3. Дефиниране на времето с брой цикли

### 3.8.3. Абсолютна форма за дефиниране на времето


- текущият момент от време се дефинира по отношение на абсолютния (нулевия) момент от време, както е показано на Фиг. 4;
- не се използва параметърът **TIMESTEP**;
- Използва се суфикс "s";

- **Пример:**

Описание на стимул сигнала от Фиг. 4

Time	Logical Level
0	0
100ns	1
200ns	0
300ns	1
400ns	1

- **Описание на стимул сигнала:** В нулевия момент сигналът е с логическо ниво 0, в момента от време 100 ns логическото ниво става 1, в момента от време 200 ns логическото ниво е 0 и т.н.



Фиг.4. Абсолютна форма на дефиниране на времето

### 3.9. 9 Команди за описание на формата на сигнала



- **<t><логическо ниво>**
- **LABEL=<име на етикет>**
- **<t> GOTO <име на етикета><n> TIMES**
- **<t> GOTO <име на етикета> UNTIL GT <данни>**
- **<t> GOTO <име на етикета> UNTIL GE <данни>**
- **<t> GOTO <име на етикета> UNTIL LT <данни>**
- **<t> GOTO <име на етикета> UNTIL LE <данни>**
- **<t> INCR BY <данни>**
- **<t> DECR BY <данни>**
- **REPEAT**
- **REPEAT FOREVER**
- **REPEAT FOR <n> TIMES**
- **ENDREPEAT**
- **FILE <име>**

- **Променлива <t>** - Моментът от време може да бъде задаван по един от трите начина на дефиниране на времето, описани в т. 3.8;
- **Променлива <данни>** - Логическите нива могат да се задават в двоична, осмична и шестнайсетична бройна система;
- При дефинирането на прехода **GOTO** управлението се предава на оператора след етикета;
- **Променлива <n>** - Определя броя на повтарящите се цикли GOTO. При задаване на **n= -1** се дефинира безкрайно повторение на цикъла;
- **Променлива <име на етикет>** - използва се за дефиниране на цикъл с помощта на оператор за преход GOTO, който предава управлението на реда, следващ оператора за етикет;
- **REPEAT FOREVER** - начало на безкрайно повтарящ се цикъл (еквивалентно на дефиниране на **n= -1**);
- **REPEAT FOR <n> TIMES** - повторение на цикъла **n** пъти;
- **ENDREPEAT** - край на цикъл, дефиниран с **REPEAT**;
- **FILE <име>** - дефинира се име на файл **<име>.stm** , в който се намира описанието на един или повече цифрови стимул сигнала.

### 3.10. Пример за описание на цифров стимул сигнал

```
USIGNAL STIM (2, 11) $G_DPWR $D_DGND
+ IN0 IN1
+ IO_STM IO_LEVEL=1
  0s      0 0
+ LABEL=LOOP
+ +100n  0 1
+ +100n  1 1
+ +100n GOTO LOOP 2 TIMES
```

- Дефиниран е цифров генератор на сигнал с два изхода в двоичен формат, включени във възли с имена съответно **IN0 IN1**, при описанието на които се използва вторият по сложност макромодел на аналого-цифровия и цифрово-аналоговия интерфейси (**IO\_LEVEL=1**);
- Използвана е относителна форма за описание на времето и е дефиниран цикъл, който се повтаря два пъти чрез оператори за преход **GOTO** и оператор за дефиниране на етикет **Label**;
- Графичният вид и резултатът от симулация на сигналите **IN0 IN1** са показани на Фиг. 5.




Фиг.5.Форма на сигнала на цифров генератор на сигнал с два изхода в двоичен формат

#### 4. Описание на цифрови стимул сигнали от Група 2, дефинирани на базата на Редактор на свойствата

Общият формат на описание на цифрови генератори на сигнал, дефинирани чрез Редактор на свойствата, е описан в точка 3. Графичното означение, свойствата и формата на запис във файла на връзките е даден за всеки сигнал поотделно.

##### 4.1. Описание на синхронизиращ (тактов) сигнал

Графичното означение:

OFFTIME = .5uS  
 ONTIME = .5uS   
 DELAY =  
 STARTVAL = 0  
 OPPVAL = 1

Свойства:

	PSpiceOnly	Reference	Value	DELAY	IO_LEVEL	IO_MODEL	OFFTIME	ONTIME	OPPVAL	Source Part	STARTVAL
DigStimuli : DigStimuli : DSTM7	TRUE	DSTM7	DigClock		0	IO_STM	.5uS	.5uS	1	DigClock.Normal	0

Запис във файла на връзките:

```

U_DSTM7 STIM(1,1) $G_DPWR $G_DGND M_UN0005
+ IO_STM IO_LEVEL=0
+ 0 0
+ .5uS 1
+REPEAT FOREVER
+ .5uS 0
+ .5uS 1
+ ENDREPEAT

```

## 4.2. Описание на единичен сигнал

Графичното означение:



Свойства:

	PSpiceOnly	Reference	Value	COMMAND1	COMMAND2	COMMAND3	COMMAND4	COMMAND5	COMMAND6	COMMAND7	
DigSimTest : DigSimTest : DSTM7	TRUE	DSTM7	STIM1	0s 0	750n 1	1500n 0					
COMMAND15	COMMAND16	DIG_GND	DIG_PWR	FORMAT	IO_LEVEL	IO_MODEL	Location X-Coordinate	Location Y-Coordinate	Source Part	TIMESTEP	WIDTH
		\$G_DGND	\$G_DPWR	1	0	IO_STM	310	380	STIM1.Normal		1

Запис във файла на връзките:

```
U_DSTM8 STIM(1,1)
+ $G_DPWR $G_DGND
+ M_UN0006
+ IO_STM
+ IO_LEVEL=0
+ 0s 0
+ 750n 1
+ 1500n 0
```

## 4.3. Описание на магистрала с 4 бита

Графичното означение:



Свойства:

	PSpiceOnly	Reference	Value	COMMAND1	COMMAND2	COMMAND3	COMMAND4	COMMAND5	COMMAND6	COMMAND7	
DigSimTest : DigSimTest : DSTM9	TRUE	DSTM9	STIM4	0s 0000	Label=loop	+100n 0100	+100n 1100	+100n goto loop 2 times			
COMMAND15	COMMAND16	DIG_GND	DIG_PWR	FORMAT	IO_LEVEL	IO_MODEL	Location X-Coordinate	Location Y-Coordinate	Source Part	TIMESTEP	WIDTH
		\$G_DGND	\$G_DPWR	1111	0	IO_STM	310	440	STIM4.Normal		4

Запис във файла на връзките:

```
U_DSTM9 STIM(4,1111)
+ $G_DPWR $G_DGND
+ M_UN0001 M_UN0002 M_UN0003 M_UN0004
+ IO_STM
+ IO_LEVEL=0
+ 0s 0000
+ Label = loop
+ +100n 0100
+ +100n 1100
+ +100n GOTO loop 2 times
```

## 4.4. Описание на магистрала с 8 бита

Графичното означение:





### Свойства:


	PSpiceOnly	Reference	Value	COMMAND1	COMMAND2	COMMAND3	COMMAND4	COMMAND5	COMMAND6	COMMAND7	COMMAND8
DigStimuli : DigStimuli : DSTM10	TRUE	DSTM10	STIM8	0s 00000000							
COMMAND15	COMMAND16	DIG_GND	DIG_PWR	FORMAT	IO_LEVEL	IO_MODEL	Location X-Coordinate	Location Y-Coordinate	Source Part	TIMESTEP	WIDTH
		\$G_DGND	\$G_DPWR	11111111	0	IO_STM	310	500	STIM8.Normal		8

### Запис във файла на връзките:

```
U_DSTM10 STIM(8,11111111)
+ $G_DPWR $G_DGND
+ M_UN0086 M_UN0087 M_UN0088 M_UN0089 M_UN0090 M_UN0091 M_UN0092 M_UN0093
+ IO_STM
+ IO_LEVEL=0
+ 0s 00000000
```

## 4.5. Описание на магистрала с 16 бита

### Графичното означение:



### Свойства:

	PSpiceOnly	Reference	Value	COMMAND1	COMMAND2	COMMAND3	COMMAND4	COMMAND5	COMMAND6	COMMAND7	
DigStimuli : DigStimuli : DSTM11	TRUE	DSTM11	STIM16	0s 0000							
COMMAND15	COMMAND16	DIG_GND	DIG_PWR	FORMAT	IO_LEVEL	IO_MODEL	Location X-Coordinate	Location Y-Coordinate	Source Part	TIMESTEP	WIDTH
		\$G_DGND	\$G_DPWR	4444	0	IO_STM	310	570	STIM16.Normal		16

### Запис във файла на връзките:

```
U_DSTM11 STIM(16,4444)
+ $G_DPWR $G_DGND
+ M_UN0007 M_UN0008 M_UN0009 M_UN0010 M_UN0011 M_UN0012 M_UN0013 M_UN0014
+ M_UN0015 M_UN0016 M_UN0017 M_UN0018 M_UN0019 M_UN0020 M_UN0021 M_UN0022
+ IO_STM
+ IO_LEVEL=0
+ 0s 0000
```

## 5. Стимул сигнали описани на базата на файл

### 5.1. Общ формат

```
1           2
Uxxx FSTIM (<брой изходи>)
           3
+ < + възел на на захранващ източник>< - възел на захр. източник>
           4
+ <списък от възли>
           5
+ <име на модела на входа/изхода>
           6
+ FILE = <име на файл.STM>
           7
+ [IO_LEVEL = <ниво на сложност на макромодела на А/Ц и Ц/А интерфейс>]
           8
+ [SIGNAME = <име на сигнала> ... ]
```

### 5.1.1. 1 Тип на цифровото устройство

**FSTIM** – служебна дума, указваща типа на цифровото устройство – цифров генератор, създаден на базата на файл.

### 5.1.2. 2 Брой изходи

<Брой изходи>

- Определя броя на изходите на цифровия генератор на сигнали.

### 5.1.3. 3 Възли на захранващия източник

- <плюс възел на захранващ източник> <минус възел на захранващ източник>;
- **\$G\_PWR** и **\$G\_DGND**.

### 5.1.4. 4 Списък от възли

- Указват се номерата на възлите, в които се включват изходите на цифровия генератор;

### 5.1.5. 5 Име на модела на входа /изхода

- **IO\_STM** - служебна дума, указваща името на модела на входа/изхода на цифровия генератор.

### 5.1.6. 6 Име на файла, в който са описани стимул сигналите

- **FILE = <име на файл.STM>**

### 5.1.7. 7 Ниво на сложност на макромодела А/Ц и Ц/А интерфейс (IO\_LEVEL)

- Незадължителен параметър;
- Дефинира нивото на сложност на макромодела на А/Ц и Ц/А интерфейс;
- 4 нива на сложност на макромодела на А/Ц и Ц/А интерфейс;
- Стойност по подразбиране - 0 (**IO\_LEVEL = 0**).

### 5.1.8. 8 Име на стимул сигналите

- Незадължителен параметър;
- Използва се служебна дума **SIGNAME**, след която се задават имената на стимул сигналите.

### 5.1.9. Пример

```
USIG FSTIM (2) $G_DPWR $G_DGND
+ IN0 IN1
+ IO_STM IO_LEVEL=0
+ FILE = "MYSTM.STM"
+ SIGNAME INPUT0 INPUT1
```

## 5.2. Структура на файла със стимул сигнали

- Позволява да бъдат използвани изходните данни от други симулатори като входни стимул сигнали;
- Представява тестов файл с <име>.stm;
- Структурата на файла се състои от заглавна част и описание на логическите състояния на сигналите;

**Заглавна част на файла със стимул сигнали**

Съдържа списък с имената на сигналите

празен ред

**Описание на логическите състояния на сигналите**

Съдържа описание на преходите на стимул сигналите

### 5.2.1. Заглавна част на файла със стимул сигнали

```
[TIMESCALE = <>]
+ <име 1>...<име n>
+ OCT ( <OCT2> < OCT1> < OCT0> )
+ HEX ( <HEX3> < HEX2> < HEX1> < HEX0> )
```

### 5.2.2. Правила за създаване на файла

- Съдържа списък от имената на сигнали;
- Между заглавната част и описанието на преходите задължително има празен ред;
- Съдържа незадължителния параметър TIMESCALE за дефиниране на мащаб по времето, който по подразбиране има стойност нула;
- Имената могат да се отделят с табулация (TAB), интервал (SPACE) и запетая;
- Имената на сигнали, записани в осмичен код, използват служебната дума **OCT( )** като се групират по три и изброяването им започва от най-старшия бит;
- Имената на сигналите, записани в шестнадесетичен код, използват служебната дума **HEX( )** като се групират по четири и изброяването им започва от най-старшия бит.

### 5.2.3. Описание на преходите на сигналите

- Задава се момент от време и след това логическото състояние;
- Възможните логическите състояния и техните означения са показани в Таблица 2, като форматът на дефиниране има следния синтаксис:

**<момент от време><логическо състояние>.**

Таблица 2. Означение на възможните логическите състояния при различните бройни системи

Логическо състояние	Двоичен формат	Осмичен формат	Шестнайсетичен формат
Логическо ниво	0,1	от 0 до 7	от 0 до F
Неопределено състояние	X	X	X
Висок импеданс	Z	Z	Z
Положителен фронт	R	R	-
Отрицателен фронт	F	F	-

### 5.2.4. Пример за дефиниране на стимул сигнал на базата на файл


```
Clock reset in1 in2
HEX (ADR7 ADR6 ADR5 ADR4)
HEX (ADR3 ADR2 ADR1 ADR0)
OCT (CTRL2 CTRL1 CTRL0)
ENABLE
(Празен ред)
0      0000  00  0  0
100n   1100  4C  2  0
200n   0101  2B  4  1
300n   1110  4C  3  1
400n   0111  FF  7  0
```

В показания пример е описана структурата на файл **StimSignal16.stm** с 16 стимул сигнала, описани както следва:

- в двоичен формат - **Clock, reset, in1, in2, ENABLE** ;
- в шестнадесетичен формат - 2 групи по 4 сигнала  
**(ADR7 ADR6 ADR5 ADR4)**  
**(ADR3 ADR2 ADR1 ADR0)**
- в осмичен формат – 1 група от 3 сигнала  
**(CTRL2 CTRL1 CTRL0).**

Описанието на сигналите във времето е абсолютна форма. Изменението на логическото състояние на стимул сигнала Clock се описва от първата колона от логически състояния, за стимул сигнала reset е втората колона от логически състояния и т.н. Например за сигнала Clock изменението на логическите състояния е следното: В нулевия момент сигналът е с логическо ниво 0, в момента от време 100 ns логическото ниво става 1, в момента от време 200 ns логическото ниво е 0 и т.н.

Логическите състояния на стимул сигналите Clock reset in1 in2 ADR7 ADR6 ADR5 ADR4 ADR3 ADR2 ADR1 ADR0 CTRL2 CTRL1 CTRL0 ENABLE в изходите на цифровия генератор на сигнали DSTIM5, които са получени след симулация с PSpice, са показани на Фиг. 6.



Фиг.6. Логически състояния на стимул сигналите в изходите на цифров генератор, описан на базата на файл **StimSignal16.stm**, които са получени след симулация с PSpice

### 5.3. Описание на единичен сигнал

Графичното означение:



FILENAME =  
SIGNAME =

Свойства:

	PSpiceOnly	Reference	Value	FILENAME	IO_LEVEL	IO_MODEL	Location X-Coordinate	Location Y-Coordinate	SIGNAME	Source Part
DigStimuli : DigStimuli : DSTM12	TRUE	DSTM12	FileStim1		0	IO_STM	480	320		FileStim1.Normal

Запис във файла на връзките:

```
U_DSTM12 FSTIM(1) $G_DPWR $G_DGND M_UN0037
IO_STM FILE="" IO_LEVEL=0
```

### 5.4. Описание на магистрала

Графичното означение:



FILENAME =  
SIGNAME =

Свойства:

	PSpiceOnly	Reference	Value	FILENAME	IO_LEVEL	IO_MODEL	Location X-Coordinate	Location Y-Coordinate	SIGNAME	Source Part
DigStimuli : DigStimuli : DSTM17	TRUE	DSTM17	FileStim32		0	IO_STM	480	630		FileStim32.Normal

Запис във файла на връзките:

```
U_DSTM17 FSTIM(32) $G_DPWR $G_DGND M_UN0038 M_UN0039 M_UN0040 M_UN0041
+ M_UN0042 M_UN0043 M_UN0044 M_UN0045 M_UN0046 M_UN0047 M_UN0048 M_UN0049
+ M_UN0050 M_UN0051 M_UN0052 M_UN0053 M_UN0054 M_UN0055 M_UN0056 M_UN0057
+ M_UN0058 M_UN0059 M_UN0060 M_UN0061 M_UN0062 M_UN0063 M_UN0064 M_UN0065
+ M_UN0066 M_UN0067 M_UN0068 M_UN0069
+ IO_STM FILE="" IO_LEVEL=0
```